SAMPLE SOLUTION FOR:

Assignment: Integer Programming Modelling with Xpress^{MP} – A Facility Location Problem

620-362 Applied Operations Research
Semester 2, 2004

Prepared by A/Prof Natashia Boland & Dr. Heng-Soon Gan

(a) **DISCUSS** why the transportation cost estimation method used above may not be suitable in reality.

Main reason:

- Euclidean or "straight-line" distances often do not well approximate distances travelled by road. In the case of Melbourne, this is especially true, with suburbs situated around the Bay. For areas on opposite sides of the Bay, in particular, such as Geelong and Region F, the connecting road distance will be seriously underestimated by using the Euclidean distance. (This might be less so if water transport is an option, but even in this case handling costs for the change of transport mode would probably outweigh any improvement.)
- "Center of region" may not be a good distance estimator for distances to demand points scattered around a region. If demand points are scattered quite evenly, and the road network sufficiently dense in the region, the approximation may be good. But if demand points are skewed or parts of the region significantly difficult to reach, for reasons such topography, infrastructure or traffic conditions, the approximation may be poor.

Other possible reasons:

- Apart from the travel distances, the transportation cost may be dependent on other infrastructure factors such as road quality or traffic conditions on the routes.
- The transportation cost will also be likely to depend on volume shipped, on whether shipments to different demand points could be combined into a single trip, etc.
- Dividing the map into regions may cause "dispute" at the region boundaries. For a given suburb in a region, it may occasionally be more economical to classify a suburb in one of its neighbouring region. This raises the question of "How should the map be partitioned?".

(b) **FORMULATE** and **SOLVE** (using Xpress^{MP}) an IP whose solution will tell Eastinghouse how to minimise the annual cost of meeting demand for air conditioners.

(c) Suppose that in fact at least 2 000 units of the region C demand for air conditioners must come from Seymour **OR** at least 2 000 units of the region C demand must come from Lilydale. Modify and re-solve your model from part (b).

(d) **(BONUS)** Using Xpress-IVE, **GENERATE A GRAPHICAL OUTPUT** of your solution. *Hint*: refer to examples provided by Xpress^{MP}. These examples can be found in C:\Program Files\XpressMP\examples. You will have to copy these examples to "D:" drive in order to run it.

The codes for parts (b), (c) and (d) are shown below. Following these codes, results in parts (b) and (c) are discussed and interpreted for the benefit of Eastinghouse, and the graphical solutions obtained shown.

```
model "FacilityLocation"

    uses "mmxprs"    !Express Optimiser Module
    uses "mmive"     !Graphical Representation Module

    parameters
        INFILE= "data.txt"
        OUTFILE= "out.txt"
    end-parameters

    forward procedure print_sol
    forward procedure draw_sol

    declarations
        SUBURB: set of string
        REGION: set of string

        f: array(SUBURB) of real         !annual fixed cost
        d: array(REGION) of real         !annual demand
        c: array(SUBURB) of real         !annual capacity
        a: array(SUBURB,REGION) of real !distance
        X,Y: array(set of string) of real    ! x-y-coordinates for graph drawing
    end-declarations

    initializations from INFILE
        f d c a X Y
    end-initializations
```

```
declarations
    x: array(SUBURB) of mpvar
    y: array(SUBURB,REGION) of mpvar
    z: mpvar
end-declarations


Cost:= sum(i in SUBURB) f(i)*x(i) + sum(i in SUBURB, j in REGION) 20*a(i,j)*y(i,j)

forall(i in SUBURB) sum(j in REGION) y(i,j) <= c(i)*x(i)
forall(j in REGION) sum(i in SUBURB) y(i,j) >= d(j)

y("SEYMOUR","C") >= 2000*z
y("LILYDALE","C") >= 2000*(1-z)

forall (i in SUBURB) x(i) is_binary
forall (i in SUBURB, j in REGION) y(i,j) is_integer
z is_binary

minimize(Cost)

print_sol

fopen(OUTFILE,F_OUTPUT)
print_sol
fclose(F_OUTPUT)

draw_sol
!-----------------------------------------------

!Procedure to print the solution
procedure print_sol
    writeln("Total Cost: $", getobjval)
    writeln

    !Write 1st row of table
    write(strfmt("Suburb ",-9))
    write(strfmt("Decision",9))
    forall(j in REGION) write(strfmt(j,9))
    writeln

    !Print roster
    forall(i in SUBURB) do
        write(strfmt(i,-9))              !Name of staff
        write(strfmt(getsol(x(i)),9))
        forall(j in REGION) do
            write(strfmt(getsol(y(i,j)),9))
        end-do
        writeln
    end-do

end-procedure
```

```
!Procedure graphical representation of the solution
procedure draw_sol
  declarations
   COLOR: array(0..6) of integer
  end-declarations

  COLOR:=[IVE_YELLOW, IVE_CYAN, IVE_RED, IVE_WHITE, IVE_GREEN, IVE_MAGENTA,
          IVE_BLACK]
  IVEerase

  FACTOR:= 1

  IVEzoom(FACTOR*-1,4,FACTOR*15,FACTOR*14)

  ct:=0
  SuburbGraph:= IVEaddplot("SUBURB", IVE_RED)
  forall(i in SUBURB) do
    IVEdrawpoint(SuburbGraph, FACTOR*X(i), FACTOR*Y(i))

    if i = "SEYMOUR" or i = "SUNBURY" then
        IVEdrawlabel(SuburbGraph, FACTOR*X(i), FACTOR*Y(i)-0.7, i)
    else
        IVEdrawlabel(SuburbGraph, FACTOR*X(i), FACTOR*Y(i)+0.1, i)
    end-if
  end-do

  RegionGraph:= IVEaddplot("REGION", IVE_BLUE)
  forall(j in REGION) do
    IVEdrawpoint(RegionGraph, FACTOR*X(j), FACTOR*Y(j))
    if X(j) <= 7 then

      forall(j in REGION) do
        IVEdrawpoint(RegionGraph, FACTOR*X(j), FACTOR*Y(j))
        if X(j) <= 7 then
            IVEdrawlabel(RegionGraph, FACTOR*X(j)-0.8, FACTOR*Y(j)-0.1, "REG " + j)
        else
            IVEdrawlabel(RegionGraph, FACTOR*X(j)+0.8, FACTOR*Y(j)-0.1, "REG " + j)
        end-if
      end-do

      ct:=0
      FlowGraph:= IVEaddplot("FLOW", COLOR(ct))
      forall(i in SUBURB) do
        if(getsol(x(i))=1) then
            forall(j in REGION) do
                if (getsol(y(i,j))<>0) then
                    IVEdrawarrow(FlowGraph, FACTOR*X(i), FACTOR*Y(i),
                    FACTOR*X(j), FACTOR*Y(j))
                    IVEdrawlabel(FlowGraph, FACTOR*X(i)/2 + FACTOR*X(j)/2,
                    FACTOR*Y(i)/2 + FACTOR*Y(j)/2, "" + getsol(y(i,j)))
                end-if
            end-do
            ct+=1
            ct:=ct MOD 7
        end-if
      end-do

    end-procedure

end-model
```

```
!Data file for FacilityLocation

!fixed cost
f: [("GEELONG") 5200000
    ("SUNBURY") 7000000
    ("HEATHCOTE") 9000000
    ("SEYMOUR") 5000000
    ("LILYDALE") 1000000
    ("FRANKSTON") 1000000]

!demand
d: [("A") 10000
    ("B") 180000
    ("C") 10000
    ("D") 180000
    ("E") 220000
    ("F") 10000]

!capacity
c: [("GEELONG") 300000
    ("SUNBURY") 150000
    ("HEATHCOTE") 220000
    ("SEYMOUR") 200000
    ("LILYDALE") 80000
    ("FRANKSTON") 80000]

!distance
a: [("GEELONG" "A") 22.56 51.06 96.19 122.31  124.69 118.75
    ("SUNBURY" "A") 57.00 41.56 46.31 67.69  81.94 95.00
    ("HEATHCOTE" "A") 100.94 67.69 11.88 55.81 99.75 131.81
    ("SEYMOUR" "A") 112.81 86.69 38.00 11.88 61.75 98.56
    ("LILYDALE" "A") 103.31 97.38 81.94 62.94 34.44 41.56
    ("FRANKSTON" "A") 83.13 90.25 99.75 96.19 72.44 54.63]

!Coordinates
X: [("GEELONG") 2.7
    ("SUNBURY") 5.4
    ("HEATHCOTE") 5.3
    ("SEYMOUR") 8.4
    ("LILYDALE") 9.9
    ("FRANKSTON") 8.1
    ("A") 1.4
    ("B") 2
    ("C") 5.3
    ("D") 9.3
    ("E") 12.3
    ("F") 12.7]

Y: [("GEELONG") 4.5
    ("SUNBURY") 8.2
    ("HEATHCOTE") 13.1
    ("SEYMOUR") 12
    ("LILYDALE") 7
```

("FRANKSTON") 4.3
("A") 5.7
("B") 8.7
("C") 12.2
("D") 12.3
("E") 8.7
("F") 5]

The output (both graphical and non-graphical) and discussions for part (b) are shown below:

The following strategies should be employed,

- Geelong should supply regions A and B with 10 000 and 90 000 units of air conditioners respectively;
- Sunbury should supply regions B, C and E with 90 000, 10 000 and 50 000 units of air conditioners respectively;
- No air conditioners are built in Heathcote;
- Seymour should supply regions D and E with 180 000 and 20 000 units of air conditioners respectively;
- Lilydale should supply region E with 80 000 units of air conditioners;
- Frankston should supply regions E and F with 70 000 and 10 000 units of air conditioners respectively;

at an annual cost of $516 544 000.

Total Cost: $5.16544e+008

Suburb	Decision	A	B	C	D	E	F
GEELONG	1	10000	90000	0	0	0	0
SUNBURY	1	0	90000	10000	0	50000	0
HEATHCOTE	0	0	0	0	0	0	0
SEYMOUR	1	0	0	0	180000	20000	0
LILYDALE	1	0	0	0	0	80000	0
FRANKSTON	1	0	0	0	0	70000	10000



The output (both graphical and non-graphical) and discussions for part (c) are shown below:

The following strategies should be employed,

- Geelong should supply regions A and B with 10 000 and 80 000 units of air conditioners respectively;
- Sunbury should supply regions B, C and E with 90 000, 8 000 and 52 000 units of air conditioners respectively;
- No air conditioners are built in Heathcote;
- Seymour should supply regions C, D and E with 2 000, 180 000 and 18 000 units of air conditioners respectively;
- Lilydale should supply region E with 80 000 units of air conditioners;
- Frankston should supply regions E and F with 70 000 and 10 000 units of air conditioners respectively;

at an annual cost of $517 019 000.

Thus the requirement that either Seymour or Lilydale supply region C with at least 2,000 units costs an additional $475 000, or 0.09%.

Total Cost: $5.17019e+008

Suburb	Decision	A	B	C	D	E	F
GEELONG	1	10000	90000	0	0	0	0
SUNBURY	1	0	90000	8000	0	52000	0
HEATHCOTE	0	0	0	0	0	0	0
SEYMOUR	1	0	0	2000	180000	18000	0
LILYDALE	1	0	0	0	0	80000	0
FRANKSTON	1	0	0	0	0	70000	10000